

Mathcad 6.0

Eine kurze Einführung in die Handhabung von Mathcad 6.0

*Abteilung Bodenphysik
Prof. Dr. B. Huwe*

Einführung in Mathcad

Vorbemerkungen

Ziele:

1. Vorstellung der Programmeigenschaften und des Leistungsumfangs
2. Einübung elementarer Arbeitsabläufe
3. Bearbeitung grundlegender mathematischer Probleme
4. Hilfe zur Selbsthilfe

Zeitplan:

1. Doppelstunde: Einführung, Erste Eindrücke, Crashkurs
2. Doppelstunde: Bearbeitung von Gleichungen und Text, Graphik
3. Doppelstunde: Numerische ingenieurmathematische Berechnungen
4. Doppelstunde: Symbolische Mathematik, Ausblick

Unterlagen:

1. Übersicht über Shortkeys
2. Übersicht über Funktionsumfang
3. Skript
4. Beispielarbeitsblätter

Was ist Mathcad

Leistungsmerkmale von Mathcad

Mathcad ist ein leistungsfähiges Mathematikprogramm, das sowohl für aufwendige und anspruchsvolle ingenieurmathematische Anwendungen als auch schulische Zwecke geeignet ist. Das besondere an Mathcad ist die spezielle Benutzerschnittstelle, die es dem Anwender erlaubt, Formeln in mathematischer Notation hinzuschreiben. Programmierkenntnisse sind daher weitgehend überflüssig.

Folgende Liste gibt einen Überblick über den Leistungsumfang von Mathcad:

- Einfache Berechnungen
- Ingenieurmathematische numerische Berechnungen
- Symbolische Mathematik
- Programmierung
- Integration von Fremdprogrammen
- Graphik
- Animationen

Erste Eindrücke

Starten Sie Mathcad durch Doppelklick auf das Mathcad-Icon. Die Mathcadoberfläche enthält viele typische Windowselemente, einige Handgriffe sind jedoch ungewohnt und müssen gesondert geübt werden.

Hilfsmittel zum Bearbeiten:

- Fadenkreuz
- Einfügemarke
- Auswahlfeld

Die Handhabung dieser Hilfsmittel wird weiter unten erläutert

Erste Berechnungsbeispiele:

- Eine **einfache Berechnung**: Klicken Sie in die freie Fläche und tippen Sie: "5.+7.="+return ein. Das Ergebnis wird sofort berechnet und angezeigt.
- Intuitive **Operatoren** sind
 - + für die Addition
 - für die Subtraktion
 - * für die Multiplikation
 - / für die Division
 - ^ für die Potenzierung
 - () für die Klammerung
 - = für die Ausführung der Berechnung
 - := für die Definition
- Ein **Integralausdruck**:
 - Klicken Sie auf die Menüleiste "Operatoren";
 - klicken Sie hier auf das Integralsymbol mit Platzhaltern für die Integrationsgrenzen;
 - klicken Sie in den Platzhalter hinter dem Integralzeichen und fügen Sie ein: "sin()dx";
 - klicken Sie in die Platzhalter für die Integrationsgrenzen und geben Sie ein "-1" (untere Grenze) und "3" (obere Grenze);
 - klicken Sie hinter dx, tippen Sie "="+return;
 das Ergebnis der numerischen Integration wird sofort angezeigt.

Übungen:

- Erkunden Sie die Mathcadoberfläche durch freies, ungeordnetes Herumklicken.
- Berechnen Sie die Porosität aus Lagerungsdichte ($= 1.25 \text{ g cm}^{-3}$) und Substanzdichte ($= 3.55 \text{ g cm}^{-3}$).
- Berechnen Sie den Wasservorrat im Boden zwischen 0 cm und 100 cm Bodentiefe, wenn sich dieser gemäß $q=0.05+0.45\cdot(1-e^{-0.5z})$ verhält.
- Spielen Sie mit den Operatoren und führen Sie auf Verdacht Berechnungen durch.

Crashkurs

Einfache Berechnungen:

Einfache Berechnungen erfolgen durch direkte Eingabe der Zahlenwerte und werden mit =↵ abgeschlossen

Beispiel:

Berechnung der Porosität aus Lagerungsdichte und Substanzdichte:

$$\left(1 - \frac{1.25}{2.65}\right) \cdot 100 = 52.83$$

Variable, Definition von Funktionen:

Die Definition von Variablen und Funktionen erfolgt mit ":=". Variablen kann hierbei eine Zahl oder ein Ausdruck aus bereits definierten Variablen zugewiesen werden. Durch letzteres wird ein Funktionsausdruck definiert.

Beispiel 1:

Verwendung von Variablen:

Substanzdichte:

$$\rho_s := 2.55$$

Lagerungsdichte:

$$\rho_b := 1.11$$

Porosität:

$$1 - \frac{\rho_b}{\rho_s} = 0.565$$

Beispiel 2:

Definition einer Funktion:

Substanzdichte:

$$\rho_s := 2.55$$

Lagerungsdichte:

$$\rho_b := 1.11$$

Porosität:

$$n := 1 - \frac{\rho_b}{\rho_s}$$

$$n = 0.565$$

(Porosität als Funktion definiert)

Text:

- Die Eingabe von Text erfolgt entweder durch Eingabe von ". Hierdurch ein Textbereich erzeugt. Anschließend kann Text eingegeben werden. Textbereiche sind frei positionierbar.

oder

- Durch Erzeugen eines Textabsatzes durch Eingabe von "Strg T". Textabsätze gehen über die ganze Seitenbreite und verhalten sich wie eine normaler Texteditor.

Iterative Berechnungen:

Iterative Berechnungen sind solche, die mehrmals hintereinander auf prinzipiell identische Art und Weise durchgeführt werden. Iterativ sind demnach z.B. Berechnungen von Feldinhalten bzw. indizierten Variablen. Hierbei beginnt Mathcad die Zählung voreingestellt bei 0. Dies kann durch Definition der Systemvariablen ORIGIN abgeändert werden.

Beispiel 1:

Berechnung der Porosität aus Lagerungsdichte und Substanzdichte für eine Serie von Meßdaten:

Erstes Feld beginnt bei mit der Zählung bei 1:

ORIGIN := 1

Definition der Bereichsvariablen i durch Eingabe von "i:1;5"

i := 1..5

Definition der Feldinhalte durch Eingabe von ":" und Eingabe der Werte durch Komma getrennt

$\rho s_i :=$ $\rho b_i :=$

2.55	1.40
2.57	1.45
2.67	1.20
2.45	1.32
2.50	1.43

Durchführung der Berechnung für jedes i

$$n_i := 1 - \frac{\rho b_i}{\rho s_i}$$

Ausgabe der Ergebnisse durch "n_i="

n_1
0.451
0.436
0.551
0.461
0.428

Beispiel 2:

Berechnung der funktionalen Abhängigkeit der Porosität von der Lagerungsdichte:

Definition der Bereichsvariablen:

$$\rho b := 0.5, 0.6.. 1.8$$

Definition der weiteren Variablen:

$$\rho s := 2.65$$

Definition der Funktion:

$$n(\rho b) := 1 - \frac{\rho b}{\rho s}$$

Ausgabe der Ergebnisse:


$n(\rho b)$
0.811
0.774
0.736
0.698
0.66
0.623
0.585
0.547
0.509
0.472
0.434
0.396
0.358
0.321

Diagramme:

Diagramme sind in Mathcad erstaunlich einfach zu erzeugen. Klicken Sie in der Graphikmemueleiste auf den gewünschten Diagrammtyp. Mathcad fügt den Graphikrumpf in Ihr Arbeitsblatt ein. Füllen Sie die Platzhalter an den Achsen mit den gewünschten Variablenbezeichnungen und Achsenbegrenzungen aus und geben Sie return ↵ ein. Die

Graphik wird sofort erzeugt. Mit Doppelklick auf die Graphik kommen Sie in das Optionenmenü für die Formatierung der Graphik. Dort können Sie die Graphik individuell ausgestalten.

Beispiel:



Eingebaute Funktionen:

Mathcad verfügt über eine Vielzahl an eingebauten Funktionen. Darüberhinaus sind eine Reihe wichtiger Konstanten vordefiniert und können mit ihrer gebräuchlichen Bezeichnung abgerufen werden.

Beispiele:

Eingebaute Grundfunktionen:

Logarithmus, Tangens, Cosinus, Sinus, usw.

$$\log(10) = 1 \quad \log(32) = 1.50515 \quad \ln(3) = 1.09861 \quad \ln(10) = 2.30259$$

$$\log(e) = 0.43429$$

$$\tan\left(\frac{\pi}{4}\right) = 1 \quad \tan\left(\frac{\pi}{8}\right) = 0.41421 \quad \sin\left(\frac{\pi}{4}\right) = 0.70711$$


Vordefinierte Konstante:

$$e = 2.71828 \quad i = i \quad \pi = 3.14159 \quad \infty = 1 \cdot 10^{307}$$

Eingebaute "höhere" Funktionen


Besselfunktion:

$$x := -20, -19.9 \dots 20$$



Fehlerfunktion:

$$x := -2, -1.9 \dots 2$$



Übungen:

- Berechnen Sie den Verlauf der Porenziffer als Funktion der Trockenraumdichte
- Stellen Sie diesen Verlauf graphisch dar.
- Berechnen Sie für die pF-Werte 0, 0.4, 0.8, 1.2, 1.6, 2.0, 2.5, 3.0, 4.2 die zugehörigen Matrixpotentiale und Äquivalentporendurchmesser.
- Analysieren Sie das Verhalten von Textbereichen und Textabsätzen

Hilfe

Mathcad verfügt wie die meisten Windowsprogramme über integrierte Hilfesysteme. Diese umfassen:

- **Hilfethemen:** Thematisch und alphabetisch abrufbare Hilfeinformationen
Hilfe erhalten Sie über den Menüpunkt "?" oder durch Drücken der Taste F1
- Hilfe zu Tastaturbefehlen bzw. -kürzeln
- Anleitung zur Benutzung der Hilfe
- Hilfe zum Technischen Kundendienst u.a. mit www- und email-Adressen
- Quicksheets (s.u.)
- Elektronische Bücher

Bem.: Sie können mit Mathcad direkt auf Informationen im Internet zugreifen. Wählen Sie hierzu den Menüpunkt Datei-URLR öffnen.

Übungen:

- Erproben Sie die Funktionsweise des Hilfesystems
- Durchforsten Sie die Inhalte der Online-Hilfe
- Studieren Sie die Informationen zur Benutzung der Tastatur
- Studieren Sie die Informationen zum Technischen Support
- Schauen Sie sich die Webseite von ***www.mathsoft.com*** an

Elektronische Bücher

Elektronische Bücher sind Sammlungen von vielen, kommentierten und funktionsfähigen Mathcad-Dokumenten zu bestimmten Themen. Sie können durch die Seiten dieser Bücher blättern, es gibt Suchfunktionen, ein Inhaltsverzeichnis sowie einen Index. Elektronische Bücher sind mittlerweile zu einer Vielzahl von Themen im Handel erhältlich. Die Inhalte der Bücher können in eigene Mathcad-Dokumente kopiert werden und sind dann ebenfalls lauffähig und editierbar. Mathcad wird mit zwei elektronischen Büchern geliefert.

Übungen:

- Studieren Sie die Funktionsweise der elektronischen Bücher.
- Durchforsten Sie die Inhalte der mitgelieferten Bücher.

QuickSheets

QuickSheets sind Anwendungsbeispiele, die als Vorlage für Eigenentwicklungen dienen können. Sie enthalten häufig Funktionsflächen zum Starten und Kopieren. Sie können innerhalb der QuickSheets experimentieren. QuickSheets können daher als Wirksame Hilfsmittel zum Lernen von Mathcad angesehen werden. Quicksheets finden Sie in der online-Hilfe.

Übungen:

- Gehen Sie in den Menüpunkt Quicksheets im Hilfemenü. Laden Sie ein Thema Ihrerer Wahl.
- Klicken Sie auf die Informationsschaltfläche und studieren Sie die Funktionsweise der Quicksheets.
- Ändern Sie die Daten in den gelben Flächen ab.
- Kopieren Sie das QuickSheet mit der Kopierschaltfläche in Ihr eigenes Areitsblatt und bringen Sie es dort zum laufen.

Bearbeiten von Gleichungen und Text

Erstellen und Bearbeiten von Gleichungen

Erstellen von Ausdrücken:

Positionieren Sie das Fadenkreuz an die gewünschte Stelle und gegen Sie die Zeichensequenz für die Formel ein.

Beispiel: Geben Sie nacheinander folgende Sequenz ein

$x-3*a^2$

/

-4+

√ aus der arithmetischen Palette

y+1

Leertaste 2x

+π in der arithmetischen Palette

Dies erzeugt folgenden Formelausdruck:

$$\frac{x-3 \cdot a^2}{-4+\sqrt{y+1}+p}$$

Klebrige Operatoren:

Divisionen, Potenzen und Wurzeln sind klebrige Operatoren, d.h. Sie passen sich der Eingabe automatisch an und werden erst durch Eingabe der Leertaste verlassen.

Auswahl eines Ausdrucks:

Erfolgt durch Drücken der -Taste oder der Leertaste

Konzept des äußeren Operators:

Der äußere Operator ist derjenige, der für den ganzen Ausdruck gilt:

$$(3+2) \cdot (5+8)$$

hier ist der Multiplikationspunkt der äußere Operator für den ganzen Ausdruck.

Tastenfunktionen:

Tasten	Funktion
	Vergrößert das Auswahlfeld Wandelt Cursor in Auswahlfeld um
↓	Verkleinert Auswahlfeld Wandelt Auswahlfeld in Cursor um
→	Verschiebt Cursor nach rechts Setzt rechts stehenden Unterausdruck in Auswahlfeld
←	Verschiebt Cursor nach links Setzt rechts stehenden Unterausdruck in Auswahlfeld
Shift	Positioniert Fadenkreuz oben außerhalb des Ausdrucks
Shift ↓	Positioniert Fadenkreuz unten außerhalb des Ausdrucks
Shift →	Positioniert Fadenkreuz rechts außerhalb des Ausdrucks
Shift ←	Positioniert Fadenkreuz links außerhalb des Ausdrucks
Leertaste	Setzt den Ausdruck eines "klebrigen" Operators in ein Auswahlfeld Ist der ganze Ausdruck im Auswahlfeld, so wird die Gleichung verlassen
Ins	Wechselt Einfügeposition auf andere Seite

Ersetzen eines Operators:

1. Schließen Sie den Ausdruck, für den der Operator äußerer Operator ist, in ein Auswahlfeld ein
2. Drücken Sie die "Del"-Taste

3. Tippen Sie neuen Operator ein

Einfügen von Klammern:

1. Schließen Sie den Ausdruck in ein Auswahlfeld ein
2. Drücken Sie die Apostroph (') - Taste

Anwenden einer Funktion auf einen Ausdruck:

1. Schließen Sie den gewünschten Ausdruck in Klammern ein (s.o.)
2. Drücken Sie die "Ins"-Taste (Das Auswahlrechteck ändert seine Farbe und die abgeschnittene Stelle erscheint links)
3. Tippen Sie den Funktionsnamen
4. Drücken Sie wieder [Ins]

Erstellen und Bearbeitungen von Text

Texte sind außerordentlich wichtig zur Kommentierung von Arbeitblättern. Die Zwei wichtigsten Techniken sind:

- Tippen von " und anschließend Eingabe von Text: hierdurch wird ein Textbereich erzeugt. Schriftart, Schriftfarbe und weitere Formatierungen können wie bei gängigen Editoren gewählt werden.
- Erzeugen eines Textabsatzes durch Eingabe von "Ctrl+T", Drücken des Symbols "¶" in der Menüleiste oder Wahl des entsprechenden Menübefehls. Textabsätze gehen über die ganze Seitenbreite und verhalten sich ebenfalls wie eine normaler Texteditor.

Gleichungen im Text einbetten:

1. Wählen Sie Berechnungsbereich einbetten aus dem Menü Text
2. Geben Sie Ihre Gleichung wie gewohnt ein
3. Kehren Sie durch Eingabe von return zum Text zurück

Griechische Buchstaben im Text:

Hierzu gibt es zwei Möglichkeiten

1. Über die Menüleiste "Griechische Buchstaben" durch Anklicken
2. Durch Eingabe des entsprechenden lateinischen Buchstabens und Drücken der Tastenkombination "Ctrl+G"

Formatieren von Gleichungen und Ergebnissen

Einstellen des globalen Ergebnisformats:

Dies erfolgt im Menü Rechnen - Numerisches Menü. Dort können folgende Angaben spezifiziert werden:

Spezifikation	Bereich	Kurzbeschreibung (näheres siehe Hilfe)
Radix	Dezimal Hexadezimal Oktal	Basis zur Darstellung der Ergebnisse
Imaginär	i, j	Buchstabe zur Anzeige imaginärer Ergebnisse
Nachfolgende Nullen	ja/nein	Auch abschließende Nullen rechts vom Komma werden angezeigt
Als Matrix anzeigen	ja/nein	"ja" unterbindet bei Matizen rollende Ausgabetafeln
Angezeigte Genauigkeit	0-15	Angezeigte (!) Anzahl der Dezimalstellen
Exponentialschwelle	0-15	Für Werte $> 10^n$ und $< 10^{-n}$ wird das Ergebnis in Exponentialschreibweise angegeben
Komplexe Toleranz	0-63	Falls das Verhältnis von Real- zu Imaginärteil $> 10^n$ oder $< 10^{-n}$ ist, wird der kleinere Teil nicht angezeigt
Nulltoleranz	0-307	Ist die Zahl $< 10^{-n}$ wird sie als Null angezeigt

Anmerkungen:

- Die interne Darstellung wird durch die oben angegebenen Formatierungen nicht beeinflusst.
- Formatierungen können lokal für eine einzige Gleichung und global für das ganze Arbeitsblatt vorgenommen werden.

Schriftarten für Berechnungen:

Konstanten, Variable, Funktionen können ebenfalls unterschiedlich formatiert werden. Dies erfolgt im Menü Rechnen - Schriftattribution.

Wichtig: Variable und Funktionen mit gleichem Inhalt aber unterschiedlicher Formatierung (Schriftart, -größe, ...) werden von Mathcad als verschiedene Variable und Funktionen interpretiert.

Konfiguration speichern

Die in 3.2 und 3.3 erläuterten Formatierungen gelten nur für das momentan bearbeitete Dokument. Sollen diese auch für andere Dokumente gelten, so müssen diese zunächst mit **Datei - Konfiguration speichern** in einer Datei abgelegt werden. In einem anderen Dokument können diese dann über **Datei - Konfigurationsdatei ausführen** aktiviert werden.

Bereiche

Formeln, Texte und Graphiken sind Bereiche, die separat ausgewählt, gelöscht, kopiert und verschoben werden können. Dies geschieht mit den üblichen Windowstechniken.

Auswahl von Bereichen:

Die Auswahl erfolgt durch Klicken außerhalb des Bereichs und Einrahmen bei gedrückter Maustaste. Zusätzliche Bereiche können durch Anklicken bei gedrückter Shift-Taste gewählt werden.

Trennen von Bereichen:

Gelegentlich überlappen sich Bereiche. Ein Trennen ist möglich von Hand oder durch Drücken von Ctrl+S. Die Ergebnisse sind manchmal etwas überraschend.

Ausrichten von Bereichen:

Bereiche können manuell und automatisch ausgerichtet werden. Beim automatischen Ausrichten werden die auszurichtenden Bereiche gewählt und anschließend der Befehl Bereiche ausrichten (horizontal oder vertikal) im Menü Bearbeiten angewandt.

Übungen:

- Üben Sie die in Kap. 3 erläuterten Techniken ein. Erstellen Sie eigene Formelausdrücke, formatieren Sie diese und stellen Sie die Ergebnisse, so möglich, graphisch dar.
- Erstellen Sie ein kommentiertes Arbeitsblatt zur Auswertung von Stechzylinderdaten. Gehen Sie von 10 Stechzylindern aus. Berechnen Sie aus Frisch-, Trocken- und Stechzylindergewicht sowie der Substanzdichte die volumtrischen und gravimetrischen Wassergehalte, Porositäten, Luftgehalte und Porenziffern der Proben.

Graphik

Überblick

Die Graphiken in Mathcad sind sehr schnell zu erstellen und decken die Palette der wissenschaftlichen Graphiken weitgehend ab. Sie haben jedoch in der Regel nicht Publikationsqualität, so daß zur Erzeugung hochwertiger Graphiken auf entsprechende Spezialsoftware zurückgegriffen werden sollte.

Folgende Graphiktypen stehen in Mathcad zur Verfügung:

- X-Y-Diagramme
- Kreisdiagramme
- Flächendiagramme
- Umrißdiagramme
- 3D-Säulendiagramme
- 3D-Streuungsdiagramme
- Vektorfelddiagramme

Darüberhinaus ist die Möglichkeit zur Erzeugung von Animationen und der Import von Bitmaps möglich. Einige Beispiele sollen nachstehend die Graphikmöglichkeiten von Mathcad illustrieren.

Die Generierung von Graphiken erfolgt in mehreren Schritten:

1. Erzeugung der Datenbasis (Dateien, Formeln)
2. Generierung eines Graphikgrundgerüsts durch klicken auf den gewünschten Graphiktyp in der Graphikmenüleiste
3. Spezifizierung der Graphikinhalte durch Eingabe der Variablen an den Platzhaltern
4. Formatierung der Graphik durch Doppelklick auf die Graphik und Selektion der gewünschten Optionen im aufgerufenen Menü.


XY-Diagramme

Beispiel für einen Scatterplot:


$$x := -5.5, -4.9 \dots 5$$

$$y(x) := 2 + 3 \cdot x + \text{rnd}(8)$$

Bem.: rnd(a) erzeugt gleichverteilte Pseudozufallszahlen zwischen 0 und a.




Beispiel: mehrere gestörte Sinusfunktionen




Contourlinien und -flächen


Beispiele für Contourlinien und Contourflächen:



3D-Diagramme

Beispiel für ein 3D-Diagramm:





Vektorfelddiagramme

Vektorfelddiagramme sind vor allem nützlich zur Veranschaulichung von Fließmustern. Hierbei müssen für jeden Punkt des Flächengitters 2 Koordinaten bereitgestellt werden. An die Graphik werden dann zwei korrespondierende Matrizen übergeben. Die eine enthält die x-Koordinaten der Pfeile, die andere die zugehörigen y-Koordinaten.

Beispiel:

```
ORIGIN := 1
```

```
i := 1 .. 20
```

```
j := 1 .. 10
```

```
xcal(x, y) := sin(2·x + y)
```


```
ycal(x, y) := cos(x2 - y)
```

```
xi := 0.15·i
```

```
yj := 0.2·j
```

```
Xi,j := xcal(xi, yj)
```

```
Yi,j := ycal(xi, yj)
```



Übungen

- Erkunden Sie die Graphikmöglichkeiten anhand der Menüpunkte der Benutzeroberfläche
- Lesen Sie die Beschreibungen in der online-Hilfe
- Studieren Sie die Graphikbeispiele in den Quicksheets
- Erstellen Sie eigene Graphiken anhand von Formeln, die Ihnen in ihrer Ausbildung begegnet sind
- Stellen Sie die funktionale Abhängigkeit der Porosität und der Poreniffer von der Lagerungsdichte und der Substanzdichte als Contourlinienplot dar
- Studieren Sie die Möglichkeiten, Animationen zu erstellen, anhand der QuickSheets

Ingenieurmathematische Berechnungen

Die nachfolgenden Ausführungen erfolgen überwiegend anhand von Beispielen und sollen einen groben Überblick über die ingenieurmathematischen Möglichkeiten von Mathcad bieten. Parallel hierzu werden am Beispiel weitere Hinweise zur Handhabung gegeben.

Konstante, Variable und Funktionen

Die Definition von Konstanten, Variablen und Funktionen erfolgt durch die Eingabe von ":". Mathcad ergänzt dies zu ":=". Auf der linken Seite von ":=" können folgende Elemente stehen:

- Einfache Variablennamen: z.B. x
- Indizierte Variablennamen: z.B. y_i

- Eine Matrix aus obigen Elementen: z.B. $\begin{pmatrix} x \\ v_i \\ z_j \end{pmatrix}$
- Eine ein- oder mehrwertige Funktion: z.B. $g(u,v,w,t)$
- Ein Zeilen oder Spaltenvektor aus einer Matrix: z.B. $\mathbf{M}^{<2>}$

Auf der rechten Seite steht dann jeweils die konkrete Definition

Beispiele:

ORIGIN := 1

Konstanten:

$a := 4$

$KI := 0.0022$

$\begin{pmatrix} r & u \\ s & v \\ t & w \end{pmatrix} := \begin{pmatrix} 2 & 33 \\ 3 & 22 \\ 55 & 99 \end{pmatrix}$ $r = 2$ $w = 99$

Variable:

$x := \begin{pmatrix} u \\ v \\ w \end{pmatrix}$ $x_1 = 33$ $x_2 = 22$ $x_3 = 99$

Dreiwertige Funktion:

$g(r,s,t) := r^2 + 5 \cdot s^t$ $g(1,2,3) = 41$

Globale Definitionen:

Globale Definitionen gelten für das gesamte Dokument, unabhängig von der Stelle an der Sie notiert wurden. Erstellt werden sie durch Eingabe der Tilde ~ anstelle des Doppelpunkts :. Mathcad wandelt die Tilde dann in das Identitätssymbol \equiv um.

Beispiel: Eingabe: R~24.123
 wird zu: R \equiv 24.123

Literalindizes:

Literalindizes sind tiefgestellte Bestandteile des Variablennamens und keine Indizes (obwohl eine optische Unterscheidung nicht möglich ist). Sie werden erzeugt durch Eingabe eines Punktes. Mathcad stellt den Text nach dem Punkt tiefer, der Punkt selbst wird nicht angezeigt.

Beispiel:

Eingabe: $\rho.b$
 wird zu ρ_b

Vordefinierte Variable

Eine Reihe von Variablenamen ist in Mathcad vordefiniert und besitzt voreingestellte Werte.

Variable	Voreinstellung	Funktion
π	3.14159...	Numerische Berechnungen: 15 Stellen Symbolische Berechnungen: exakter Wert von π
e	2.71828...	Numerische Berechnungen: 15 Stellen Symbolische Berechnungen: exakter Wert von e
∞	10^{307}	Ist in Berechnungen ein endlicher Wert!!
%	0.01	Als Einheit verwendet
TOL	10^{-3}	Konvergenzkriterium für Iterationsalgorithmen
ORIGIN	0	Index des ersten Elements eines Feldes
PRNCOLWIDTH	8	Spaltenbreite die zum Schreiben von Dateien mit PRNSCHREIBEN verwendet wird
PRNPRECISION	4	Anzahl der signifikanten Stellen, die beim Schreiben von Dateien mit PRNSCHREIBEN verwendet werden
FRAME	0	Zur Ausführung von Animationen. Keine Animation: FRAME=0

Vektoren, Matrizen

Das **Erstellen von Vektoren** erfolgt mit Hilfe der Menüleiste "Vektoren- und Matrizenoperatoren auf sehr einfache Art und Weise. In diesem Menü sind auch die wichtigsten Operatoren für Vektoren zu finden nachfolgend sind einige Beispiele hierzu aufgelistet.

ORIGIN := 1

Definition der Vektoren:

$$\mathbf{u} := \begin{pmatrix} 1 \\ 4 \\ 2 \end{pmatrix} \quad \mathbf{v} := \begin{pmatrix} 3 \\ 2 \\ 4.5 \end{pmatrix} \quad \mathbf{w} := \begin{pmatrix} 2.1 \\ 3 \\ 12 \end{pmatrix}$$

Ansprache einzelner Vektorelemente:

$$u_2 = 4 \quad w_3 = 12$$

Elementare Vektoroperationen:

Vektorsumme: $\mathbf{u} + \mathbf{w} = \begin{pmatrix} 3.1 \\ 7 \\ 14 \end{pmatrix}$

Multiplikation mit einem Skalar: $5 \cdot \mathbf{v} = \begin{pmatrix} 15 \\ 10 \\ 22.5 \end{pmatrix}$

Skalarprodukt: $\mathbf{u} \cdot \mathbf{v} = 20$

Vektorprodukt: $\mathbf{u} \times \mathbf{v} = \begin{pmatrix} 14 \\ 1.5 \\ -10 \end{pmatrix}$

Vektorsumme: $\sum \mathbf{u} = 7 \quad \sum \mathbf{v} = 9.5 \quad \sum \mathbf{w} = 17.1$

$$\sum (\mathbf{u} + \mathbf{v} + \mathbf{w}) = 33.6$$

Das **Erzeugen von Matrizen** geschieht im Prinzip auf die gleiche Art und Weise. Vektoren und Matrizen können jedoch auch berechnet oder aus Dateien eingelesen werden. Eine weitere Möglichkeit ist das Zusammensetzen einer Matrix aus einzelnen Vektoren:

$$\mathbf{u} := \begin{pmatrix} 1 \\ 4 \\ 2 \end{pmatrix} \quad \mathbf{v} := \begin{pmatrix} 3 \\ 2 \\ 4.5 \end{pmatrix} \quad \mathbf{w} := \begin{pmatrix} 2.1 \\ 3 \\ 12 \end{pmatrix}$$

$$A := \text{erweitern}(u, v) \quad A = \begin{pmatrix} 1 & 3 \\ 4 & 2 \\ 2 & 4.5 \end{pmatrix} \quad B := \text{erweitern}(A, w) \quad B = \begin{pmatrix} 1 & 3 & 2.1 \\ 4 & 2 & 3 \\ 2 & 4.5 & 12 \end{pmatrix}$$

Elementare Matrixoperationen:

Transponieren, Herausgreifen einer Spalte:

$$B^T = \begin{pmatrix} 1 & 4 & 2 \\ 3 & 2 & 4.5 \\ 2.1 & 3 & 12 \end{pmatrix}$$

$$B^{<1>} = \begin{pmatrix} 1 \\ 4 \\ 2 \end{pmatrix}$$

$$B^{<1>T} = (1 \ 4 \ 2)$$

$$(B^T)^{<2>} = \begin{pmatrix} 4 \\ 2 \\ 3 \end{pmatrix}$$

Multiplikation von Matrizen mit Skalaren,
Vektoren und Matrizen:

$$C := 5 \cdot B$$

$$C = \begin{pmatrix} 5 & 15 & 10.5 \\ 20 & 10 & 15 \\ 10 & 22.5 & 60 \end{pmatrix}$$

$$z := B \cdot v$$

$$z = \begin{pmatrix} 18.45 \\ 29.5 \\ 69 \end{pmatrix}$$

$$B \cdot B = \begin{pmatrix} 17.2 & 18.45 & 36.3 \\ 18 & 29.5 & 50.4 \\ 44 & 69 & 161.7 \end{pmatrix}$$

$$B \cdot C = \begin{pmatrix} 86 & 92.25 & 181.5 \\ 90 & 147.5 & 252 \\ 220 & 345 & 808.5 \end{pmatrix}$$

**Produkt einer Matrix mit ihrer
Transponierten:**

$$B \cdot B^T = \begin{pmatrix} 14.41 & 16.3 & 40.7 \\ 16.3 & 29 & 53 \\ 40.7 & 53 & 168.25 \end{pmatrix}$$

Inverse einer Matrix:

$$B^{-1} = \begin{pmatrix} -0.122 & 0.308 & -0.056 \\ 0.488 & -0.091 & -0.063 \\ -0.163 & -0.017 & 0.116 \end{pmatrix}$$

$$B \cdot B^{-1} = \begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Determinante einer Matrix:

$$d := |B|$$

$$d = -86.1$$

Anwendungsbeispiel: Hauptachsentransformation:

Ausgangsmatrix:

$$S := \begin{pmatrix} 2 & 3 & 6 \\ 3 & 4 & 7 \\ 6 & 7 & 2 \end{pmatrix}$$

1. Bestimmung der Eigenwerte:

$v := \text{eigenwerte}(S)$

$$v = \begin{pmatrix} -0.109 \\ -5.409 \\ 13.518 \end{pmatrix}$$

2. Bestimmung der Eigenvektoren:

$ev0 := \text{eigenvek}(S, v_0)$

$ev1 := \text{eigenvek}(S, v_1)$

$ev2 := \text{eigenvek}(S, v_2)$

$$ev0 = \begin{pmatrix} -0.747 \\ 0.661 \\ -0.068 \end{pmatrix}$$

$$ev1 = \begin{pmatrix} -0.454 \\ -0.434 \\ 0.778 \end{pmatrix}$$

$$ev2 = \begin{pmatrix} 0.485 \\ 0.612 \\ 0.625 \end{pmatrix}$$

Bem: Eigenvektoren stehen senkrecht aufeinander:

$$ev0 \cdot ev1 = 0$$

$$ev0 \cdot ev2 = 0$$

$$ev1 \cdot ev2 = 0$$

3. Bestimmung der Transformationsmatrix T:

$T := \text{erweitern}(ev0, ev1)$

$T := \text{erweitern}(T, ev2)$

$$T = \begin{pmatrix} -0.747 & -0.454 & 0.485 \\ 0.661 & -0.434 & 0.612 \\ -0.068 & 0.778 & 0.625 \end{pmatrix}$$

Überprüfung:
$$T^{-1} \cdot S \cdot T = \begin{pmatrix} -0.109 & 0 & 0 \\ 0 & -5.409 & 0 \\ 0 & 0 & 13.518 \end{pmatrix}$$

Bem.: Auf die Möglichkeit, Operationen zu vektorisieren, d.h. Operationen auf Vektoren und Matrizen parallel abzuarbeiten, wird hier nur hingewiesen. Einzelheiten hierzu sind in der online-Hilfe zu finden.

Rekursive Iteration

Rekursive Iterationen kommen in der numerischen Mathematik sehr häufig vor. Wichtige Anwendungen sind numerische Verfahren zur Lösung partieller Differentialgleichungen wie zum Beispiel Diffusionsprobleme. Das nachstehende Beispiel zeigt das Prinzip des Verfahrens anhand der Berechnung einer Quadratwurzel. Trotz schlechter Anfangsschätzung konvergiert das gewählte Verfahren außerordentlich schnell.

Geschätzt werden soll eine Näherung für $\sqrt{2800} = 52.915$ durch eine rekursive Methode.

Problemdefinition:

$$\begin{aligned} N &:= 10 \\ i &:= 0..N \\ R &:= 2800 \end{aligned}$$

Anfangsschätzung:

$$W_0 := 1$$

Algorithmus:

$$W_{i+1} := \left(W_i + \frac{R}{W_i} \right) \cdot \frac{1}{2}$$

i	W_i	$(W_i)^2$
0	1	1
1	$1.401 \cdot 10^3$	$1.961 \cdot 10^6$
2	701.25	491751.062
3	352.621	124341.751
4	180.281	32501.201
5	97.906	9585.606
6	63.252	4000.875
7	53.76	2890.112
8	52.922	2800.702
9	52.915	2800
10	52.915	2800

Interpolieren

Eine häufige Anwendung bei der Auswertung von Daten ist die Interpolation. Hierzu gibt es wiederum eine Reihe von Möglichkeiten, die durchaus nicht alle zum gleichen Ergebnis führen. Eine Möglichkeit, Daten glatt zu interpolieren bieten die spline-Funktionen. Auch existieren verschiedene Varianten. Bei der Interpretation solcher Ergebnisse, insbesondere an den Rändern des Datenbereichs ist daher grundsätzlich Vorsicht geboten.

ORIGIN := 1

$x :=$	$\begin{bmatrix} -7 \\ -3 \\ -1 \\ 1 \\ 3 \\ 5 \\ 7 \\ 9 \\ 11 \\ 18 \end{bmatrix}$	$y :=$	$\begin{bmatrix} 8 \\ 6 \\ 22 \\ 35 \\ 4 \\ -8 \\ -12 \\ 12 \\ 8 \\ 12 \end{bmatrix}$
--------	---	--------	---

$s := -12, -11.9 .. 19 \quad i := 1 .. 10$

lineare Interpolation:


$$g(s) := \text{linterp}(x, y, s)$$

verschiedene spline-Interpolationen:

$$vs := \text{lspline}(x, y) \quad h(s) := \text{interp}(vs, x, y, s)$$

$$us := \text{pspline}(x, y) \quad r(s) := \text{interp}(us, x, y, s)$$

$$ws := \text{kspline}(x, y) \quad t(s) := \text{interp}(ws, x, y, s)$$



Integrieren, Differenzieren

Besonders bei Integrationen zeigt sich, wie einfach Mathcad zu handhaben ist. Ohne weitere Programmierarbeiten sind auch umfangreiche Integrationen einfach zu bewältigen.

Integrationen:

$$i := 0..5$$

einfaches Integral:

$$f(i, x) := e^x + (i \cdot x)^2$$

$$h_1 := \int_{-2}^2 f(i, x) dx$$

h_i

7.254
12.587
28.587
55.254
92.587
140.587

Doppelintegral:

$$\int_{-5}^5 \int_{-2}^3 \sin(x-y)^2 \cdot (x \cdot y) \, dx \, dy = 2.051$$

Berechnung des Schwerpunkts eines Dreiecks mit einer Massendichte proportional zum Ursprung:

$$d(x, y) := \sqrt{x^2 + y^2}$$

$$\text{Masse} := \int_0^1 \int_0^x d(x, y) \, dy \, dx \quad \text{Masse} = 0.383$$

$$x_{\text{mittel}} := \frac{1}{\text{Masse}} \cdot \int_0^1 \int_0^x x \cdot d(x, y) \, dy \, dx \quad x_{\text{mittel}} = 0.75$$

$$y_{\text{mittel}} := \frac{1}{\text{Masse}} \cdot \int_0^1 \int_0^x y \cdot d(x, y) \, dy \, dx \quad y_{\text{mittel}} = 0.398$$

Ähnlich einfach sind Ableitungen erster und höherer Ordnung zu bewältigen. Ist die Ableitung auf analytische Art zu ermitteln, so ist dies häufig vorzuziehen. Das nachstehende Beispiel zeigt die Anwendung der numerischen Differentiation anhand einer einfachen Funktion.

Definition der Funktion:

$$x := 0, 0.1 \dots 10$$


$$f(x) := 0.01 \cdot x^3 + 2 \cdot x^2 + 10 \cdot \sin(2 \cdot x)$$

Erste, zweite und dritte Ableitung:

$$df(x) := \frac{d}{dx} f(x)$$

$$d^2f(x) := \frac{d^2}{dx^2} f(x)$$

$$d^3f(x) := \frac{d^3}{dx^3} f(x)$$

***Lösen von Gleichungen***

Mathcad verfügt über umfangreiche Möglichkeiten zum Lösen von Gleichungen mit und ohne Nebenbedingungen. Hierzu gehört auch das Anpassen nichtlinearer Funktionen. Das folgende Beispiel zeigt die Anpassung einer analytischen Funktion an gemessene Diffusionskoeffizienten als Funktion der luftgefüllten Porosität. Zur Anwendung kommt das Levenberg-Marquardt-Verfahren.

ORIGIN := 1

L := 5

m := 1..L

i := 1..5

z := 5

$D_i :=$ $\text{poros}_i :=$

0.015	0.169
0.0295	0.179
0.063	0.209
0.079	0.239
0.0865	0.28

Anfitten der Kurve:

Definition des Modells: $S(a, b, c, \text{poros}) := a \cdot e^{-\frac{b}{\text{poros} - c}}$

Optimalitätskriterium: $SSE(a, b, c) := \sum_i (D_i - S(a, b, c, \text{poros}_i))^2$

Festlegung der Anfangsschätzungen:

$$a := 0.115 \quad c := 0.153 \quad b := 0.034$$

Definition der Bedingungen:

given
 $SSE(a, b, c) = 0 \quad 1 = 1 \quad 1 = 1$
 (Pro Parameter eine Gleichung)

Aufruf der Optimierungsprozedur:

$$\begin{pmatrix} a \\ b \\ c \end{pmatrix} := \text{minfehl}(a, b, c)$$


Lösungsvektor:

$$\begin{pmatrix} a \\ b \\ c \end{pmatrix} = \begin{pmatrix} 0.115 \\ 0.034 \\ 0.153 \end{pmatrix}$$

Visualisierung:

$$z := 0.16, 0.161 \dots 0.3$$

$$\text{fit}(z) := S(a, b, c, z)$$



Einlesen von Daten, deskriptive Statistik

Das Einlesen von Daten erfolgt überwiegend in Form rechteckiger Datenmatrizes. Das Einlesen erfolgt über den Befehl PRNLESEN(fn). Der Parameter fn ist hierbei der interne filename. Die Verknüpfung mit der externen Datei erfolgt über Datei-Dateinamen verknüpfen. Dort sind die internen Dateinamen aufgelistet und können dem externen file durch Selektion zugeordnet werden.

Beispiel: Exploration eines Datensatzes:

Dateneingabe (Datenmatrix M):

```
M := PRNLESEN(abc)
```


Selektieren der einzelnen Datenspalten:

```
dist := M<0>      Cu := M<1>      Co := M<2>      Ni := M<3>      Cr := M<4>
```

Visualisierung:

```
i := 1 .. 101
```

Zusammenhang zwischen den Variablen:



Deskriptive Statistik:

mittelwert(Cu) = 19.99

var(Cu) = 35.912

stdabw(Cu) = 5.993

mittelwert(Cr) = 99.676

var(Cr) = $4.288 \cdot 10^3$

stdabw(Cr) = 65.486

mittelwert(Co) = 18.627

var(Co) = 61.508

stdabw(Co) = 7.843

mittelwert(Ni) = 167.794

var(Ni) = $1.082 \cdot 10^4$

stdabw(Ni) = 104.011

$\text{korr}(\text{Cu}, \text{Co}) = 0.377$	$\text{korr}(\text{Cu}, \text{Ni}) = 0.397$	$\text{korr}(\text{Cu}, \text{Cr}) = 0.347$
	$\text{korr}(\text{Co}, \text{Ni}) = 0.921$	$\text{korr}(\text{Co}, \text{Cr}) = 0.78$
		$\text{korr}(\text{Ni}, \text{Cr}) = 0.71$

Symbolische Berechnungen

Überblick

Ähnlich wie in MAPLE oder MATHEMATIKA sind auch in Mathcad symbolische Berechnungen möglich. Im Unterschied zu numerischen sind symbolische Berechnungen i.a. exakte mathematische Manipulationen. Die in Mathcad möglichen symbolischen Operationen umfassen hierbei:

- Algebraische Manipulation von Ausdrücken
- Symbolische Auswertung unbestimmter Integrale, Ableitungen und Grenzwerte
- Algebraisches Lösen von Gleichungen und Gleichungssystemen
- Ermittlung symbolischer Transponierter, Inverser und Determinanten von Matrizen
- Symbolische Fourier-, Laplace- und z-Transformationen
- Symbolische Vereinfachung komplexer Gleichungen

Nachstehen werden einige Beispiele zur symbolischen Berechnung angegeben, die die Möglichkeiten des symbolischen Prozessors jedoch bestenfalls andeuten können.

Symbolische Differential- und Integralrechnung, Symbolische Berechnung von Grenzwerten, Partialbruchzerlegung

Symbolisches bestimmtes Integral:

Bem.: das symbolische Gleichheitszeichen wird erzeugt durch die Eingabe von "**Ctrl . F9**"

$$\int_a^b x^2 dx \Rightarrow \frac{1}{3} \cdot b^3 - \frac{1}{3} \cdot a^3$$

$$\int_u^v \sin(x)^2 \cdot \tan(x) dx \Rightarrow \frac{-1}{2} \cdot \sin(v)^2 - \ln(\cos(v)) + \frac{1}{2} \cdot \sin(u)^2 + \ln(\cos(u))$$

Symbolisches unbestimmtes Integral:

$$\int \sin(x) + x \, dx \rightarrow \frac{1}{2} \cdot x^2 - \cos(x)$$

$$\int \ln(x) \cdot \cos(x)^2 \, dx \rightarrow \frac{1}{4} \cdot \ln(x) \cdot \sin(2 \cdot x) + \frac{1}{2} \cdot x \cdot \ln(x) - \frac{1}{2} \cdot x - \frac{1}{4} \cdot \text{Si}(2 \cdot x) + \frac{1}{8} \cdot \pi \cdot \text{csgn}(x)$$

Bem.: a) $\text{csgn}(x) = 1$, wenn $\text{Re}(x) > 0$ bzw. wenn $\text{Re}(x) = 0$ und $\text{Im}(x) \geq 0$, sonst $\text{csgn}(x) = -1$

$$\text{b) Si}(x) = \int_0^x \frac{\sin(t)}{t} \, dt$$

Symbolisch Differenzieren:

$$\frac{\sin(x)^2 \cdot \tan(x)}{23 + x^3}$$

differenziert, ergibt symbolisch:

$$2 \cdot \sin(x) \cdot \frac{\tan(x)}{(23 + x^3)} \cdot \cos(x) + \sin(x)^2 \cdot \frac{(1 + \tan(x)^2)}{(23 + x^3)} - 3 \cdot \sin(x)^2 \cdot \frac{\tan(x)}{(23 + x^3)^2} \cdot x^2$$

Bem.: Variable x markieren, dann Befehl

Partialbruchzerlegung aus dem Menü symbolisch wählen

Grenzwerte:

$$\lim_{x \rightarrow a} \frac{3 \cdot x + b}{x^2} \quad \text{ergibt symbolisch:} \quad \frac{(3 \cdot a + b)}{a^2}$$

Partialbruchzerlegung:

$$\frac{2 \cdot x^2 - 3 \cdot x + 1}{x^3 + 2 \cdot x^2 - 9 \cdot x - 18} \quad \text{ergibt symbolisch:} \quad \frac{1}{(3 \cdot (x - 3))} + \frac{14}{(3 \cdot (x + 3))} - \frac{3}{(x + 2)}$$

Symbolisches Lösen von Gleichungen

Beispiel 1:

$$e^x = -1$$

nach x aufgelöst ergibt:

$$i \cdot \pi$$

Bem.: das Zeichen ■ = ■ wird erzeugt durch Eingabe von **Ctrl +**

Beispiel 2:

$$\frac{1}{2} \cdot x^2 + x + 2 = 0$$

hat als Lösungen

$$\begin{pmatrix} -1 + i \cdot \sqrt{3} \\ -1 - i \cdot \sqrt{3} \end{pmatrix}$$

Beispiel 3:

$$\sin(x) = \frac{1}{7} \cdot \tan(x)$$

hat als Lösungen

$$\begin{pmatrix} 0 \\ \arcsin\left(\frac{1}{7}\right) \end{pmatrix}$$

Beispiel 4: Lösen einer Ungleichung

$$(x^3 - 5 \cdot x^2 - 4 \cdot x) + 20 > 0$$

hat als Lösungen:

$$\left[\begin{array}{l} (-2 < x) \cdot (x < 2) \\ 5 < x \end{array} \right]$$

Bem.: Das Symbol "." ist hier als logisches "und" zu interpretieren

Symbolische Matrixoperationen

Symbolische Transponierung:

$$\begin{pmatrix} x & f & 8 \\ 2 & 5 & u \\ z & 6 & v \end{pmatrix} \quad \text{symbolisch transponiert, ergibt:} \quad \begin{pmatrix} x & 2 & z \\ f & 5 & 6 \\ 8 & u & v \end{pmatrix}$$

Symbolische Determinante:

$$\begin{pmatrix} x & f & 8 \\ 2 & 5 & u \\ z & 6 & v \end{pmatrix} \text{ besitzt die Determinante: } 5 \cdot x \cdot v - 6 \cdot x \cdot u - 2 \cdot f \cdot v + 96 + z \cdot f \cdot u - 40 \cdot z$$

Symbolische Inverse:

$$\begin{pmatrix} x & f & 8 \\ 2 & 5 & u \\ z & 6 & v \end{pmatrix}$$

hat die Inverse:

$$\frac{1}{(5 \cdot x \cdot v - 6 \cdot x \cdot u - 2 \cdot f \cdot v + 96 + z \cdot f \cdot u - 40 \cdot z)} \begin{pmatrix} 5 \cdot v - 6 \cdot u & -f \cdot v + 48 & f \cdot u - 40 \\ -2 \cdot v + u \cdot z & x \cdot v - 8 \cdot z & -x \cdot u + 16 \\ 12 - 5 \cdot z & -6 \cdot x + f \cdot z & 5 \cdot x - 2 \cdot f \end{pmatrix}$$

Kontrolle:

$$\begin{pmatrix} x & f & 8 \\ 2 & 5 & u \\ z & 6 & v \end{pmatrix} \cdot \frac{1}{(5 \cdot x \cdot v - 6 \cdot x \cdot u - 2 \cdot f \cdot v + 96 + z \cdot f \cdot u - 40 \cdot z)} \begin{pmatrix} 5 \cdot v - 6 \cdot u & -f \cdot v + 48 & f \cdot u - 40 \\ -2 \cdot v + u \cdot z & x \cdot v - 8 \cdot z & -x \cdot u + 16 \\ 12 - 5 \cdot z & -6 \cdot x + f \cdot z & 5 \cdot x - 2 \cdot f \end{pmatrix}$$

symbolisch multipliziert, ergibt:

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

Literatur

MathSoft (1995): Mathcad - Benutzerhandbuch Mathcad 6.0/Mathcad PLUS 6.0. MathSoft, Cambridge.